



Summer Fellowship Report

On

Custom Modelling of Thermodynamic Packages in
OpenModelica

Submitted by

S. Aakash Subramanian

Under the guidance of

Prof.Kannan M. Moudgalya
Chemical Engineering Department
IIT Bombay

July 23, 2018

Acknowledgment

I wish to express my deepest gratitude to my internship guide Dr. Kannan M. Moudgalya, Professor, Department of Chemical Engineering, IIT Bombay for his continuous support and supervision throughout the internship. I have reaped benefits from his wisdom, guidance and patience all along.

I would also like to express my profound and heart felt thanks to my professor Dr. P. R. Naren, SCBT, SASTRA Deemed University, A. S. Rahul, Department of Chemical Engineering, IIT Bombay, Priyam Nayak, Department of Chemical Engineering, IIT Bombay and the FOSSEE team for their timely help and guidance.

Contents

1	Introduction	3
2	Background	4
3	Soave - Redlich - Kwong (SRK) Equation of State	5
3.1	Introduction	5
3.2	Equation	5
3.3	Nomenclature	6
3.4	Input	7
3.5	OpenModelica Code	7
3.6	Results	11
4	Chao - Seader	12
4.1	Introduction	12
4.2	Equation	12
4.3	Nomenclature	13
4.4	Input	14
4.5	OpenModelica Code	15
4.6	Output	18

Chapter 1

Introduction

“OpenModelica” is a free and open-source modelling environment that uses “Modelica” modelling language. It follows equation oriented approach. OpenModelica can be used for modelling, simulation, optimization and analysis of complex steady state and dynamic systems. Modelica modelling language allows users to express a system in the form of equations. OpenModelica compiles expressions, equations, functions and algorithms into efficient C code. The generated C code is combined with a library of utility functions, a run-time library, and a numerical Differential-Algebraic Equation (DAE) solver. OpenModelica Connection Editor, called as OMEdit is the integrated Graphical User Interface (GUI) in OpenModelica for graphical modelling and editing. OMEdit consists of several libraries for various domains like Electrical, Magnetic, Math, Thermal, etc. It provides various user friendly features like representation of a model in the form of block diagrams. OMEdit can be used for creating custom models and for editing or drawing connections between the model interfaces. It also allows users to plot graphs between parameters of the model simulated.

Chapter 2

Background

OMEdit, the integrated Graphical User Interface (GUI) of OpenModelica consists of libraries across various domains such as Electrical, Thermal, Math, Mechanics, etc. that can be used to develop models of a physical system or process. However, OMEdit does not contain libraries for modelling of chemical process systems. This limits the use of OpenModelica for modelling chemical process systems. Hence, it becomes necessary to build models or packages that aids in modelling of chemical process systems. An essential requirement is the availability of thermodynamic packages to estimate fluid properties. The main aim of this work is to develop standard thermodynamic packages in OpenModelica. In this work, two standard thermodynamic packages, namely “Soave - Redlich - Kwong” (SRK) and “Chao - Seader” are developed in OpenModelica.

Inorder to develop thermodynamic packages in OpenModelica, properties such as critical temperature, critical pressure, acentric factor, molecular weight, etc. of the components that constitutes the system are required. A port has already been created from DWSIM to OpenModelica so that all the required properties are obtained from ChemSep Database present in DWSIM. The developed thermodynamic packages can be used to calculate fluid pure component and solution properties such as equilibrium constants, molar composition, residual enthalpy and residual entropy. These packages will serve as the backbone for developing custom models of any chemical process systems.

Chapter 3

Soave - Redlich - Kwong (SRK) Equation of State

3.1 Introduction

In 1972, G. Soave proposed a modification to the Redlich - Kwong (RK) Equation of State and derived a new equation of state called as “Soave - Redlich - Kwong”. Soave - Redlich - Kwong (SRK) Equation of State is an empirical and algebraic equation that represents the state of a system. It is an equation relating molar volume and temperature of a system to its pressure. It is generally used for systems consisting of light hydrocarbon gases such as methane, ethane, propane, etc. It can also model the Vapour - Liquid Equilibria (VLE) properties of fluids, like fugacity coefficients, molar compositions of components in both liquid and vapour phases, equilibrium constants, enthalpy, entropy of the system and so on, in addition to the vapour phase properties

3.2 Equation

Soave - Redlich - Kwong (SRK) Equation of State:

$$P = \frac{RT}{V - b} + \frac{a(T)}{V(V + b)} \quad (3.1)$$

$$a(T) = (1 + (0.48 + 1.574\omega - 0.176\omega^2)(1 - \sqrt{T_r}))^2 \left[\frac{0.42747(R^2 T_c^2)}{P_c} \right] \quad (3.2)$$

$$b = 0.08664 \frac{RT_c}{P_c} \quad (3.3)$$

For mixtures consisting of more than one component

$$a_m = \sum_i^n \sum_j^n ((x_i x_j \sqrt{a_i a_j})(1 - k_{ij})) \quad (3.4)$$

$$b_m = \sum_i^n x_i b_i \quad (3.5)$$

Soave - Redlich - Kwong (SRK) Equation of State based on Compressibility Factor:

$$Z^3 - Z^2 + (A - B - B^2)Z - AB = 0 \quad (3.6)$$

$$A = \frac{a_m P}{R^2 T^2} \quad (3.7)$$

$$B = \frac{b_m P}{RT} \quad (3.8)$$

$$Z = \frac{PV}{RT} \quad (3.9)$$

Fugacity Coefficient of component in mixture:

$$\ln \phi_i = \frac{b_i(Z-1)}{b_m} - \ln(Z-B) - \left(\frac{A}{B} \frac{2 \sum_j^n x_j \sqrt{a_j a_i} (1 - k_{ji})}{a_m} - \frac{b_i}{b_m} \right) \ln \left(\frac{Z+B}{Z} \right) \quad (3.10)$$

Equilibrium Constant of component:

$$K_i = \frac{\phi_{i,l}}{\phi_{i,v}} \quad (3.11)$$

$$y_i = K_i x_i \quad (3.12)$$

$$y_i \beta + x_i (1 - \beta) = z_i \quad (3.13)$$

$$\sum_i^n y_i - \sum_i^n x_i = 0 \quad (3.14)$$

3.3 Nomenclature

$a(T)$ – Soave - Redlich - Kwong Equation of State constant of component

a_m – Soave - Redlich - Kwong Equation of State constant of mixture

A – Soave - Redlich - Kwong Equation of State constant of mixture

b – Soave - Redlich - Kwong Equation of State constant of component

b_m – Soave - Redlich - Kwong Equation of State constant of mixture

B – Soave - Redlich - Kwong Equation of State constant of mixture

i – Component

k_{ij} – Interaction Parameter between i^{th} component and j^{th} component

K – Equilibrium constant of component

l – Liquid

n – Total number of components

P – Pressure of the System, Pa

P_c – Critical Pressure of component, Pa

P_r – Reduced Pressure

R – Universal Gas Constant

$R = 8.314 \frac{kJ}{kmolK}$

T – Temperature of the System, K

T_c – Critical Temperature of component, K

T_r – Reduced Temperature

v – Vapour

V – Molar Volume of the System, $\frac{m^3}{mol}$
 x_i – Molar fraction of i^{th} component in liquid phase
 y_i – Molar fraction of i^{th} component in vapour phase
 β – Molar fraction of feed vapourized
 ω – Acentric Factor of component
 ϕ – Fugacity Coefficient of component

3.4 Input

Components

1. Propylene
2. Ethylene

Thermodynamic Package

Soave - Redlich - Kwong (SRK)

Stream Specification

Table 3.1: Input

Parameter	Units	Feed
Temperature	K	200
Pressure	Pa	101325
Molar Flow	$\frac{mol}{s}$	100
Composition	-	
Propylene	-	0.5
Ethylene	-	0.5

3.5 OpenModelica Code

```

//Soave - Redlich - Kwong Equation Of State
//Components
//Propylene - 1
//Ethylene - 2
//Liquid - l
//Vapour - v
//=====
//Aakash Subramanian S.
//SASTRA Deemed University
//=====
model SRK
  
```



```

//=====
//Parameter Section

parameter Integer NOC = 2 "Number of Components";

parameter Real R(unit = "J/(mol K)") = 8.314 "Universal Gas Constants";

parameter Real MW[NOC](unit = "kg/kgmol") = {42.0797, 28.054} "Molecular Weight";

parameter Real z[NOC] = {0.5, 0.5} "Molar Composition of Feed";

parameter Real Tc[NOC](unit = "K") = {364.8, 282.34} "Critical Temperature";

parameter Real Pc[NOC](unit = "Pa") = {4600000, 5041000} "Critical Pressure";

parameter Real omega[NOC] = {0.137588, 0.087} "Acentric Factor";

parameter Real uu = 1, ww = 0;

//=====
//Variable Section

input Real T(unit = "K", start = 200) "Temperature";
input Real P(unit = "Pa", start = 101325) "Pressure";

Real a[NOC](each start = 1) "SRK EOS Constants";
Real b[NOC](each start = 0) "SRK EOS Constants";

Real aml2[NOC], bml2[NOC], amv2[NOC], bmv2[NOC];
Real aml "SRK EOS Constant for Liquid Mixture";
Real bml "SRK EOS Constant for Liquid Mixture";

Real amv "SRK EOS Constant for Vapour Mixture";
Real bmv "SRK EOS Constant for Vapour Mixture";

Real Al "SRK EOS Constant for Liquid Mixture";
Real Bl "SRK EOS Constant for Liquid Mixture";

Real Av "SRK EOS Constant for Vapour Mixture";
Real Bv "SRK EOS Constant for Vapour Mixture";

Real Zl(start = 0) "Compressibility Factor of Liquid";
Real Zv(start = 1) "Compressibility Factor of Vapour";

Real x[NOC](each start = 1, each min = 0, each max = 1) "Mole Fraction in Liquid Phase";
Real y[NOC](each start = 1, each min = 0, each max = 1) "Mole Fraction in Vapour Phase";

Real r[NOC];
Real s[NOC];

Real t1[NOC], t2, t3[NOC], t4, t5;
Real u1[NOC], u2, u3[NOC], u4, u5;

Real beta(start = 0) "Mole Fraction of Feed Vapourized";

Real phil[NOC](each start = 1) "Fugacity Coefficient in Liquid Phase";
Real phiv[NOC](each start = 1) "Fugacity Coefficient in Vapour Phase";

Real K[NOC](each start = 1) "Equilibrium Constant";

Real c[NOC];
Real a1, al, av, dadT[2], a12[NOC], av2[NOC];
Real AMWL(unit = "kg/kgmol") "Average Molecular Weight of Liquid Mixture";
Real AMWv(unit = "kg/kgmol") "Average Molecular Weight of Vapour Mixture";
Real mDHres[2](unit = "kJ/kmol") "Residual Molar Enthalpy";
Real mDSres[2](unit = "kJ/(kmol K)") "Residual Molar Entropy";
Real DHres[2](unit = "kJ/kg") "Residual Specific Enthalpy";
Real DSres[2](unit = "kJ/(kg K)") "Residual Specific Entropy";

```

```

Real DAres[2](unit = "kJ/kmol") "Residual Specific Helmholtz Free Energy";

//=====
//Equation Section

equation

//Calculation of SRK EOS Constants for Liquid and Vapour Phases

aml = sum(aml2);
bml = sum(bml2);

Al = (aml*P)/((R*T)^2);
Bl = (bml*P)/(R*T);

(Zl^3)-(Zl^2)+((Al-Bl-(Bl^2))*Zl)-(Al*Bl) = 0; //Compressibility Factor of
Liquid Mixture

amv = sum(amv2);
bmv = sum(bmv2);

Av = (amv*P)/((R*T)^2);
Bv = (bmv*P)/(R*T);

(Zv^3)-(Zv^2)+((Av-Bv-(Bv^2))*Zv)-(Av*Bv) = 0; //Compressibility Factor of
Vapour Mixture

t2 = -log(Zl-Bl);
t4 = log((Zl+Bl)/Zl);
t5 = Bl;

u2 = -log(Zv-Bv);
u4 = log((Zv+Bv)/Zv);
u5 = Bv;

sum(y)-sum(x) = 0;

for i in 1:NOC loop
  t1[i] = b[i]*(Zl-1)/bml;
  t3[i] = Al*((2*r[i]/aml)-(b[i]/bml));

  u1[i] = b[i]*(Zv-1)/bmv;
  u3[i] = Av*((2*s[i]/amv)-(b[i]/bmv));

//Calculation of Fugacity Coefficients of Components in Liquid and Vapour Phases

  phil[i] = exp(t1[i]+t2-(t3[i]*t4/t5));
  phiv[i] = exp(u1[i]+u2-(u3[i]*u4/u5));

//Calculation of Equilibrium Constants

  K[i] = phil[i]/phiv[i];

//Calculation of Mole Fraction of Components in Liquid and Vapour Phases

  y[i] = K[i]*x[i];
  z[i] = (y[i]*beta) + (x[i]*(1-beta));

//Calculation of SRK EOS Constants of Components

  a[i] = ((1+((0.48+(1.574*omega[i])-(0.176*(omega[i]^2)))*(1-((T/Tc[i])^0.5))))^2)
    *((0.42747*(R*R*Tc[i]*Tc[i])/Pc[i]));
  b[i] = 0.08664*R*Tc[i]/Pc[i];

  aml2[i] = sum(x[i].*x[:].*((a[i].*a[:]).^0.5));
  bml2[i] = (x[i]*b[i]);

  amv2[i] = sum(y[i]*y[:].*((a[i].*a[:]).^0.5));
  bmv2[i] = (y[i]*b[i]);

```

```

r[i] = sum(x[:].*((a[:].*a[i]).^0.5));
s[i] = sum(y[:].*((a[:].*a[i]).^0.5));

c[i] = 0.48 + 1.574 * omega[i] - 0.176 * omega[i]^2;
av2[i] = sum(y[i].*y[:].*(c[:].*(a[i].*Tc[:]./Pc[:]).^0.5+c[i].*(a[:].*Tc[i]./Pc[i]).^0.5));
al2[i] = sum(x[i].*x[:].*(c[:].*(a[i].*Tc[:]./Pc[:]).^0.5+c[i].*(a[:].*Tc[i]./Pc[i]).^0.5));
end for;

//Calculation of Average Molecular Weights of Liquid and Vapour Phases

AMWv = sum(y[:].*MW[:]);
AMWl = sum(x[:].*MW[:]);

//Calculation of Intermediate Values

a1 = -R / 2 * (0.42748 / T) ^ 0.5;

a1 = sum(a1);
av = sum(av2);

dadT[1] = a1*a1;
dadT[2] = a1*av;

//Calculation of Residual Properties of Liquid Phase

DAres[1] = aml/(bml*(uu^2-4*ww)^0.5)*log((2*Zl+B1*(uu-(uu^2-4*ww)^0.5))/(2*Zl+B1*(uu+(uu^2-4*ww)^0.5)))-R*T*log((Zl-B1)/Zl)-R*T*log(Zl);
mDSres[1] = R*log((Zl-B1)/Zl)+R*log(Zl)-uu/((uu^2-4*ww)^0.5*bml)*dadT[1]*log((2*Zl+B1*(uu-(uu^2-4*ww)^0.5))/(2*Zl+B1*(uu+(uu^2-4*ww)^0.5)));
mDHres[1] = DAres[1]+T*mDSres[1]+R*T*(Zl-1);
DHres[1] = mDHres[1]/AMWl;
DSres[1] = mDSres[1]/AMWl;

//Calculation of Residual Properties of Vapour Phases

DAres[2] = amv/bmv*log((2*Zv+Bv*(uu-(uu^2-4*ww)^0.5))/(2*Zv+Bv*(uu+(uu^2-4*ww)^0.5)))-R*T*log((Zv-Bv)/Zv)-R*T*log(Zv);
mDSres[2] = R*log((Zv-Bv)/Zv)+R*log(Zv)-uu/((uu^2-4*ww)^0.5*bmv)*dadT[2]*log((2*Zv+Bv*(uu-(uu^2-4*ww)^0.5))/(2*Zv+Bv*(uu+(uu^2-4*ww)^0.5)));
mDHres[2] = DAres[2]+T*mDSres[2]+R*T*(Zv-1);
DHres[2] = mDHres[2]/AMWv;
DSres[2] = mDSres[2]/AMWv;

end SRK;
=====

```

3.6 Results

Table 3.2: Output

Parameter	Units	OpenModelica	DWSIM	Percent Error
Phase - Liquid				
Composition				
Propylene	-	0.82185	0.82173	0.02
Ethylene	-	0.17815	0.17826	0.06
Residual Molar Enthalpy	$\frac{kJ}{kmol}$	-18416.5	-18419.1	0.01
Residual Molar Entropy	$\frac{kJ}{kmolK}$	-85.2018	-85.2053	0.004
Fugacity Coefficients				
Propylene	-	0.266877	0.266400	0.18
Ethylene	-	4.25659	4.25630	0.006
Phase - Vapour				
Propylene	-	0.227859	0.227423	0.19
Ethylene	-	0.772141	0.772577	0.06
Residual Molar Enthalpy	$\frac{kJ}{kmol}$	-100.381	-100.356	0.03
Residual Molar Entropy	$\frac{kJ}{kmolK}$	-0.313648	-0.313572	0.02
Fugacity Coefficients				
Propylene	-	0.962581	0.962572	0.0009
Ethylene	-	0.982091	0.982090	0.0001
Equilibrium Constants				
Propylene	-	0.277252	0.276759	0.18
Ethylene	-	4.33421	4.33392	0.007

Chapter 4

Chao - Seader

4.1 Introduction

Chao - Seader method uses Redlich - Kwong (RK) Equation of State for calculation of thermodynamic properties such as fugacity coefficient, molar composition, molar volume, etc. Other thermodynamic properties such as residual molar enthalpy and residual molar entropy of a system are calculated using Lee - Kesler - Plocker (LKP) Equation of State.

4.2 Equation

Chao - Seader uses Redlich - Kwong (RK) Equation of State to calculate constants.

Redlich - Kwong (RK) Equation of State:

$$P = \frac{RT}{V - b} - \frac{a}{\sqrt{T}V(V + b)} \quad (4.1)$$

$$a = 0.42748 \frac{R^2 T_c^{2.5}}{P_c T^{0.5}} \quad (4.2)$$

$$b = 0.08664 \frac{RT_c}{P_c} \quad (4.3)$$

For mixtures containing more than one component:

$$a_m = \sum_i^n \sum_j^n x_i x_j a_{ij} \quad (4.4)$$

$$b_m = \sum_i^n x_i b_i \quad (4.5)$$

Fugacity coefficient of component in mixture

$$\ln \phi_i = b_i \frac{Z-1}{b_m} - \log(Z-B) + \frac{A}{B} \left(\frac{b_i}{b_m} - 2 \sqrt{\frac{a_i}{a_m}} \right) \log\left(\frac{Z+B}{Z}\right) \quad (4.6)$$

Equilibrium Constant of component:

$$K_i = \frac{\phi_{i,l}}{\phi_{i,v}} \quad (4.7)$$

$$y_i = K_i x_i \quad (4.8)$$

$$y_i \beta + x_i (1 - \beta) = z_i \quad (4.9)$$

$$\sum_i^n y_i - \sum_i^n x_i = 0 \quad (4.10)$$

4.3 Nomenclature

a – Redlich - Kwong Equation of State constant of component

a_m – Redlich - Kwong Equation of State constant of mixture

A – Redlich - Kwong Equation of State constant of mixture

b – Redlich - Kwong Equation of State constant of component

b_m – Redlich - Kwong Equation of State constant of mixture

B – Redlich - Kwong Equation of State constant of mixture

i – Component

K – Equilibrium constant of component

l – Liquid

n – Total number of components

P – Pressure of the system, Pa

P_c - Critical Pressure of component, Pa

P_r – Reduced Pressure

R – Universal Gas Constant

$$R = 8.314 \frac{kJ}{kmolK}$$

T – Temperature of the System, K
 T_c – Critical Temperature of component, K
 T_r – Reduced Temperature
 v – Vapour
 V – Molar Volume of the system, $\frac{m^3}{mol}$
 x_i - Mole Fraction of i^{th} component in liquid phase
 y_i - Mole Fraction of i^{th} component in vapour phase
 β – Molar fraction of feed vapourized
 ω – Acentric Factor of component
 ϕ – Fugacity Coefficient of component

4.4 Input

Components

1. Ethane
2. Propane

Thermodynamic Package

Chao - Seader

Stream Specification

Table 4.1: Input

Parameter	Units	Feed
Temperature	K	215
Pressure	Pa	101325
Molar Flow	$\frac{mol}{s}$	100
Composition	-	
Ethane	-	0.5
Propane	-	0.5

4.5 OpenModelica Code

```
//Chao – Seader Equation of State
//Components
//Ethane – 1
//Propane – 2
//Liquid – l
//Vapour – v
//=====
//Aakash Subramanian S.
//SASTRA Deemed University
//=====
model cs

//=====
//Parameter Section

parameter Integer NOC = 2 "Number of Components";
parameter Real R(unit = "J/(mol K)") = 8.314 "Universal Gas Constant";
parameter Real z[NOC] = {0.5, 0.5} "Molar Composition of Feed";
parameter Real Tc[NOC](unit = "K") = {305.32, 369.83} "Critical Temperature";
parameter Real Pc[NOC](unit = "Pa") = {4872000, 4248000} "Critical Pressure";
parameter Real omega[NOC] = {0.099, 0.152} "Acentric Factor";
parameter Real comega[NOC] = {0.0908000, 0.153800} "Chao – Seader Acentric Factor";
parameter Real sp[NOC] = {12400, 13100} "Solubility Parameter";
parameter Real u = 1, w = 0;

//Actiity Coefficient Parameters

parameter Real f0 = 5.75748;
parameter Real f1 = -3.01761;
parameter Real f2 = -4.985;
parameter Real f3 = 2.02299;
parameter Real f4 = 0;
parameter Real f5 = 0.08427;
parameter Real f6 = 0.26667;
parameter Real f7 = -0.31138;
parameter Real f8 = -0.02655;
parameter Real f9 = 0.02883;

//=====
//Variable Section

input Real T(unit = "K", start = 215) "Temperature of System";
input Real P(unit = "Pa", start = 101325) "Pressure of System";

Real Tr[NOC] "Reduced Temperature";
Real Pr[NOC] "Reduced Pressure";

Real beta(start = 0) "Mole Fraction of Feed Vapourised";

Real x[NOC](each start = 0, each min = 0, each max = 1) "Mole Fraction in Liquid Phase";
Real y[NOC](each start = 0, each min = 0, each max = 1) "Mole Fraction in Vapour Phase";

Real a[NOC] "SRK EOS Constants";
Real b[NOC] "SRK EOS Constants";

Real amv2[NOC], bmv2[NOC];
```



```

Real amv "SRK EOS Constant for Vapour Mixture";
Real bmv "SRK EOS Constant for Vapour Mixture";
Real Av "SRK EOS Constant for Vapour Mixture";
Real Bv "SRK EOS Constant for Vapour Mixture";

Real Zv(start = 1) "Compressibility Factor of Vapour";

Real t1[NOC], t2, t3[NOC], t4;

Real phil[NOC] "Fugacity Coefficients in Liquid Phase";
Real phiv[NOC] "Fugacity Coefficients in Vapour Phase";

Real K[NOC] "Equilibrium Constants";

Real gamma[NOC] "Activity Coefficients";

Real sumV, sumVS, S;

Real V[NOC] "Partial Volume";

Real v0[NOC], v1[NOC], v[NOC];

//=====
//Equation Section
equation
for i in 1:NOC loop
    Tr[i] = T/Tc[i]; //Reduced Temperature Calculation
    Pr[i] = P/Pc[i]; //Reduced Pressure Calculation

//Calculation of SRK EOS Constants
    a[i] = (0.42748*(R^2)*(Tc[i]^2.5))/(Pc[i]*(T^0.5));
    b[i] = 0.08664*R*Tc[i]/Pc[i];

//Vapour Phase Fugacity Coefficient Calculations
    amv2[i] = sum((y[i].*y[:].*((a[i].*a[:]).^0.5)));
    bmv2[i] = (y[i]*b[i]);

    t1[i] = b[i]*(Zv-1)/bmv;
    t3[i] = Av/(Bv*(u^2-4*w)^0.5)*(b[i]/bmv-2*(a[i]/amv)^0.5);

    phiv[i] = exp(t1[i]+t2+t3[i]*t4);

//Liquid Phase Fugacity Coefficient Calculations
//Calculation of Activiy Coefficients of Liquid
    gamma[i] = exp(V[i]*((sp[i]-S)^2)/(8314470.0*T));

    v0[i] = f0 + f1 / Tr[i] + f2 * Tr[i] + f3 * Tr[i]^2 + f4 * Tr[i]^3 + (f5 +
        f6 * Tr[i] + f7 * Tr[i]^2) * Pr[i] + (f8 + f9 * Tr[i]) * Pr[i]^2 - log10(
        Pr[i]);
    v1[i] = -4.23893 + 8.65808 * Tr[i] - 1.2206 / Tr[i] - 3.15224 * Tr[i]^3 - 0.025
        * (Pr[i] - 0.6);
    log10(v[i]) = v0[i]+(comega[i]*v1[i]);

//Calculations of Equilibrium Constants
    phil[i] = gamma[i]*v[i];

    K[i] = phil[i]/phiv[i];

```

```

//Calculation of Mole Fraction of Components in Liquid and Vapour Phases

y[i] = K[i]*x[i];
z[i] = (y[i]*beta) + (x[i]*(1-beta));

end for;

//Partial Volume of Components in Mixture

V[1] = 0.0590406;
V[2] = 0.0733157;

//Calculation of SRK EOS Constants for Vapour Mixture

amv = sum(amv2);
bmv = sum(bmv2);

//Calculation of SRK EOS Constants for Vapour Mixture

Av = (amv*P)/((R*T)^2);
Bv = (bmv*P)/(R*T);

//Calculation of Compressibility Factor of Vapour Mixture

(Zv^3)-((1+Bv-(u*Bv))*(Zv^2))+((Av+(w*(Bv^2))-(u*Bv)-(u*(Bv^2)))*Zv)-((Av*Bv)+(w*(Bv^2))+w*(Bv^3))) = 0;

t2 = -log(Zv-Bv);
t4 = log((2*Zv+Bv*(u+(u^2-4*w)^0.5))/(2*Zv+Bv*(u-(u^2-4*w)^0.5)));

//Calculation of Intermediate Parameters

sumV = sum(x[:].*V[:]);
sumVS = sum(x[:].*V[:].*sp[:]);
S = sumVS/sumV;

sum(y)-sum(x) = 0;
end cs;
=====

```

4.6 Output

Table 4.2: Output

Parameter	Units	OpenModelica	DWSIM	Percent Error
Phase - Liquid				
Composition				
Ethane	-	0.155204	0.1552065	0.002
Propane	-	0.844796	0.844793	0.0004
Fugacity Coefficients				
Ethane	-	3.64852	3.648473	0
Propane	-	0.483432	0.4834323	0
Phase - Vapour				
Composition				
Ethane	-	0.576795	0.576796	0.0001
Propane	-	0.423205	0.423204	0.0002
Fugacity Coefficients				
Ethane	-	0.981745	0.98174515	0
Propane	-	0.965021	0.96502084	0
Equilibrium Constants				
Ethane	-	3.71636	3.716313	0
Propane	-	0.500955	0.500955	0